

4 Compositional Semantics and Algebraic Structures


```
{-# LANGUAGE FlexibleInstances, GeneralizedNewtypeDeriving #-}
module DSLsofMath.W04 where
import Prelude hiding (Monoid)
import DSLsofMath.FunExp
```

4.1 Compositional semantics

4.1.1 A simpler example of a non-compositional function

Consider a very simple datatype of integer expressions:

```
data E = Add E E | Mul E E | Con Integer deriving Eq
e1, e2 :: E
e1 = Add (Con 1) (Mul (Con 2) (Con 3)) -- 1 + 2 * 3
e2 = Mul (Add (Con 1) (Con 2)) (Con 3) -- (1 + 2) * 3
```



When working with expressions it is often useful to have a “pretty-printer” to convert the abstract syntax trees to strings like “1+2*3”.

```
pretty :: E → String
```

We can view *pretty* as an alternative *eval* function for a semantics using *String* as the semantic domain instead of the more natural *Integer*. We can implement *pretty* in the usual way as a “fold” over the syntax tree using one “semantic constructor” for each syntactic constructor:

```
pretty (Add x y) = prettyAdd (pretty x) (pretty y)
pretty (Mul x y) = prettyMul (pretty x) (pretty y)
pretty (Con c)   = prettyCon c
prettyAdd :: String → String → String
prettyMul :: String → String → String
prettyCon :: Integer → String
```

Now, if we try to implement the semantic constructors without thinking too much we would get the following:

```
prettyAdd xs ys = xs ++ "+" ++ ys
prettyMul xs ys = xs ++ "*" ++ ys
prettyCon c     = show c
p1, p2 :: String
p1 = pretty e1
p2 = pretty e2
```

```
trouble :: Bool
trouble = p1 == p2
```

Note that both e_1 and e_2 are different but they pretty-print to the same string. This means that the parsing (the inverse of *pretty*) is not a homomorphism. There are many ways to fix this, some more “pretty” than others, but the main problem is that some information is lost in the translation.

For the curious. One solution to the problem with parentheses is to create three (slightly) different functions intended for printing in different contexts. The first of them is for the top level, the second for use inside *Add*, and the third for use inside *Mul*. These three functions all have type $E \rightarrow \text{String}$ and can thus be combined with the tupling transform into one function returning a triple: $\text{prVersions} :: E \rightarrow (\text{String}, \text{String}, \text{String})$. The result is the following:

```
prTop :: E → String
prTop e = let (pTop, -, -) = prVersions e
           in pTop

type ThreeVersions = (String, String, String)
prVersions :: E → ThreeVersions
prVersions = foldE prVerAdd prVerMul prVerCon

prVerAdd :: ThreeVersions → ThreeVersions → ThreeVersions
prVerAdd (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInA ++ "+" ++ yInA    -- use InA because we are “in Add”
  in (s, paren s, paren s)      -- parens needed except at top level

prVerMul :: ThreeVersions → ThreeVersions → ThreeVersions
prVerMul (xTop, xInA, xInM) (yTop, yInA, yInM) =
  let s = xInM ++ "*" ++ yInM    -- use InM because we are “in Mul”
  in (s, s, paren s)            -- parens only needed inside Mul


prVerCon :: Integer → ThreeVersions
prVerCon i =
  let s = show i
  in (s, s, s)                  -- parens never needed

paren :: String → String
paren s = "(" ++ s ++ ")"
```

Exercise: Another way to make this example go through is to refine the semantic domain from *String* to $\text{Precedence} \rightarrow \text{String}$. This can be seen as another variant of the result after the tupling transform: if *Precedence* is an n -element type then $\text{Precedence} \rightarrow \text{String}$ can be seen as an n -tuple. In our case a three-element *Precedence* would be enough.

4.1.2 Compositional semantics in general

In general, for a syntax *Syn*, and a possible semantics (a type *Sem* and an *eval* function of type $\text{Syn} \rightarrow \text{Sem}$), we call the semantics *compositional* if we can implement *eval* as a fold. Informally a “fold” is a recursive function which replaces each abstract syntax constructor C_i of *Syn* with a “semantic constructor” ci . Thus, in our datatype *E*, a compositional semantics means that *Add* maps to *add*, *Mul* \mapsto *mul*, and *Con* \mapsto *con* for some “semantic functions” *add*, *mul*, and *con*.



As an example we can define a general *foldE* for the integer expressions:

```
foldE :: (s -> s -> s) -> (s -> s -> s) -> (Integer -> s) -> (E -> s)
foldE add mul con = rec
  where rec (Add x y) = add (rec x) (rec y)
        rec (Mul x y) = mul (rec x) (rec y)
        rec (Con i)   = con i
```

Notice that *foldE* has three function arguments corresponding to the three constructors of *E*. The “natural” evaluator to integers is then easy:

```
evalE1 :: E -> Integer
evalE1 = foldE (+) (*) id
```

and with a minimal modification we can also make it work for other numeric types:

```
evalE2 :: Num a => E -> a
evalE2 = foldE (+) (*) fromInteger
```

Another thing worth noting is that if we replace each abstract syntax constructor with itself we get the identity function (a “deep copy”):

```
idE :: E -> E
idE = foldE Add Mul Con
```

Finally, it is often useful to capture the semantic functions (the parameters to the fold) in a type class:

```
class IntExp t where
  add :: t -> t -> t
  mul :: t -> t -> t
  con :: Integer -> t
```

In this way we can “hide” the arguments to the fold:

```
foldIE :: IntExp t => E -> t
foldIE = foldE add mul con

instance IntExp E where
  add = Add
  mul = Mul
  con = Con

instance IntExp Integer where
  add = (+)
  mul = (*)
  con = id

idE' :: E -> E
idE' = foldIE

evalE' :: E -> Integer
evalE' = foldIE
```

To get a more concrete feeling for this, we define some concrete values, not just functions:

```
seven :: IntExp a => a
seven = add (con 3) (con 4)
```

```

testI :: Integer
testI = seven
testE :: E
testE = seven
check :: Bool
check = and [ testI == 7
              , testE == Add (Con 3) (Con 4)
              , testP == "3+4"
            ]

```

We can also see *String* and *pretty* as an instance:

```

instance IntExp String where
  add = prettyAdd
  mul = prettyMul
  con = prettyCon
  pretty' :: E → String
  pretty' = foldIE
  testP :: String
  testP = seven


```

To sum up, by defining a class *IntExp* (and some instances) we can use the methods (*add*, *mul*, *con*) of the class as “smart constructors” which adapt to the context. An overloaded expression, like *seven :: Num a ⇒ a*, which only uses these smart constructors can be instantiated to different types, ranging from the syntax tree type *E* to different semantic interpretations (like *Integer*, and *String*).

4.1.3 Back to derivatives and evaluation

Review section 3.9 again with the definition of *eval'* being non-compositional (just like *pretty*) and *evalD* a more complex, but compositional, semantics.

We want to implement $eval' = eval \circ derive$ in the following diagram:



As we saw in section 3.9 this does not work in the sense that *eval'* cannot directly be implemented compositionally. The problem is that some of the rules of computing the derivative depends not only on the derivative of the subexpressions, but also on the subexpressions before taking the derivative. A typical example of the problem is *derive* (*f* *∗* *g*) where the result involves not only *derive* *f* and *derive* *g*, but also *f* and *g*.

The solution is to extend the return type of *eval'* from one semantic value *f* of type *Func* = $\mathbb{R} \rightarrow \mathbb{R}$ to two such values $(f, f') :: (Func, Func)$ where $f' = D f$. One way of expressing this is to say that in order to implement $eval' :: FunExp \rightarrow Func$ we need to also compute $eval :: FunExp \rightarrow Func$. Thus we need to implement a pair of *eval*-functions (*eval*, *eval'*) together. Using the “tupling transform” we can express this as computing just one function $evalD :: FunExp \rightarrow (Func, Func)$ returning a pair of *f* and *D f* at once.

This combination *is* compositional, and we can then get *eval'* back as the second component of *evalD e*:

```
eval' :: FunExp → Func
eval' = snd ∘ evalD
```

4.2 Algebraic Structures and DSLs

In this section, we continue exploring the relationship between type classes, mathematical structures, and DSLs.

4.2.1 Algebras, homomorphisms

The mathematical theory behind compositionality talks about homomorphisms between algebraic structures.

From Wikipedia:

In universal algebra, an algebra (or algebraic structure) is a set *A* together with a collection of operations on *A*.

Example:

```
class Monoid a where
  unit :: a
  op    :: a → a → a
```

After the operations have been specified, the nature of the algebra can be further limited by axioms, which in universal algebra often take the form of identities, or *equational laws*.

Example: Monoid equations

A monoid is an algebra which has an associative operation 'op' and a unit. The laws can be formulated as the following equations:

$$\begin{aligned} \forall x : a. (unit \text{ 'op' } x == x \wedge x \text{ 'op' } unit == x) \\ \forall x, y, z : a. (x \text{ 'op' } (y \text{ 'op' } z) == (x \text{ 'op' } y) \text{ 'op' } z) \end{aligned}$$

Examples of monoids include numbers with additions, $(\mathbb{R}, 0, (+))$, numbers with multiplication $(\mathbb{R}_{>0}, 1, (*))$, and even endofunctions with composition $(a \rightarrow a, id, (\circ))$. It is a good exercise to check that the laws are satisfied. (An “endofunction” is simply a function of type $X \rightarrow X$ for some set X .)

In mathematics, as soon as there are several examples of a structure, the question of what “translation” between them comes up. An important class of such “translations” are “structure preserving maps” called *homomorphisms*. As two examples, we have the homomorphisms *exp* and *log*, specified as follows:

```
exp : ℝ      → ℝ>0
exp 0       = 1      -- e0 = 1
exp (a + b) = exp a * exp b -- ea+b = eaeb
log : ℝ>0 → ℝ
```

$$\begin{aligned} \log 1 &= 0 && \text{-- } \log 1 = 0 \\ \log (a * b) &= \log a + \log b && \text{-- } \log(ab) = \log a + \log b \end{aligned}$$

What we recognize as the familiar laws of exponentiation and logarithms are actually examples of the homomorphism conditions for *exp* and *log*. Back to Wikipedia:

More formally, a homomorphism between two algebras A and B is a function $h: A \rightarrow B$ from the set A to the set B such that, for every operation f_A of A and corresponding f_B of B (of arity, say, n), $h(f_A(x_1, \dots, x_n)) = f_B(h(x_1), \dots, h(x_n))$.

Our examples *exp* and *log* are homomorphisms between monoids and the general monoid homomorphism conditions for $h: A \rightarrow B$ are:

$$\begin{aligned} h \text{ unit} &= \text{unit} && \text{-- } h \text{ takes units to units} \\ h(x \text{ 'op' } y) &= h x \text{ 'op' } h y && \text{-- and distributes over op (for all } x \text{ and } y) \end{aligned}$$

Note that both *unit* and *op* have different types on the left and right hand sides. On the left they belong to the monoid $(A, \text{unit}_A, \text{op}_A)$ and on the right they belong to $(B, \text{unit}_B, \text{op}_B)$.

To make this a bit more concrete, here are two examples of monoids in Haskell: the additive monoid *ANat* and the multiplicative monoid *MNat*.

```
newtype ANat    = A Int deriving (Show, Num, Eq)
instance Monoid ANat where
    unit          = A 0
    op (A m) (A n) = A (m + n)
newtype MNat    = M Int deriving (Show, Num, Eq)
instance Monoid MNat where
    unit          = M 1
    op (M m) (M n) = M (m * n)
```

In mathematical texts the constructors M and A are usually omitted and below we will stick to that tradition.

Exercise: characterise the homomorphisms from *ANat* to *MNat*.

Solution:

Let $h: \text{ANat} \rightarrow \text{MNat}$ be a homomorphism. Then it must satisfy the following conditions:

$$\begin{aligned} h 0 &= 1 \\ h(x + y) &= h x * h y && \text{-- for all } x \text{ and } y \end{aligned}$$

For example $h(x + x) = h x * h x = (h x)^2$ which for $x = 1$ means that $h 2 = h(1 + 1) = (h 1)^2$.

More generally, every n in *ANat* is equal to the sum of n ones: $1 + 1 + \dots + 1$. Therefore

$$h n = (h 1)^n$$

Every choice of $h 1$ “induces a homomorphism”. This means that the value of the function h for any natural number, is fully determined by its value for 1.

Exercise: show that *const* is a homomorphism. The distribution law can be shown as follows:

$$\begin{aligned} h a + h b &= \{- h = \text{const in this case -}\} \\ \text{const } a + \text{const } b &= \{- \text{By def. of } (+) \text{ on functions -}\} \\ (\lambda x \rightarrow \text{const } a x + \text{const } b x) &= \{- \text{By def. of } \text{const}, \text{ twice -}\} \end{aligned}$$

$$\begin{array}{ll}
(\lambda x \rightarrow a + b) & = \{- \text{By def. of } \mathit{const} -\} \\
\mathit{const} (a + b) & = \{- \mathit{h} = \mathit{const} -\} \\
\mathit{h} (a + b) &
\end{array}$$

We now have a homomorphism from values to functions, and you may wonder if there is a homomorphism in the other direction. The answer is “Yes, many”. Exercise: Show that *apply* *c* is a homomorphism for all *c*, where *apply* *x f* = *f x*.

4.2.2 Homomorphism and compositional semantics

Earlier, we saw that *eval* is compositional, while *eval'* is not. Another way of phrasing that is to say that *eval* is a homomorphism, while *eval'* is not. To see this, we need to make explicit the structure of *FunExp*:

```

instance Num FunExp where
  (+)      = (:+ :)
  (*)      = (:* :)
  fromInteger = Const o fromInteger
instance Fractional FunExp where
  -- Exercise: fill in
instance Floating FunExp where
  exp      = Exp
  -- Exercise: fill in

```

and so on.

Exercise: complete the type instances for *FunExp*.

For instance, we have

$$\begin{array}{l}
\mathit{eval} (e_1 :* e_2) = \mathit{eval} e_1 * \mathit{eval} e_2 \\
\mathit{eval} (\mathit{Exp} e) = \mathit{exp} (\mathit{eval} e)
\end{array}$$

These properties do not hold for *eval'*, but do hold for *evalD*.

The numerical classes in Haskell do not fully do justice to the structure of expressions, for example, they do not contain an identity operation, which is needed to translate *Id*, nor an embedding of doubles, etc. If they did, then we could have evaluated expressions more abstractly:

$$\mathit{eval} :: \mathit{GoodClass} a \Rightarrow \mathit{FunExp} \rightarrow a$$

where *GoodClass* gives exactly the structure we need for the translation. With this class in place we can define generic expressions using smart constructors just like in the case of *IntExp* above. For example, we could define

$$\begin{array}{l}
\mathit{twoexp} :: \mathit{GoodClass} a \Rightarrow a \\
\mathit{twoexp} = \mathit{mulF} (\mathit{constF} 2) (\mathit{expF} \mathit{idF})
\end{array}$$

and instantiate it to either syntax or semantics:

$$\begin{array}{l}
\mathit{testFE} :: \mathit{FunExp} \\
\mathit{testFE} = \mathit{twoexp} \\
\mathit{testFu} :: \mathit{Func} \\
\mathit{testFu} = \mathit{twoexp}
\end{array}$$

Exercise: define the class *GoodClass* and instances for *FunExp* and $\text{Func} = \mathbb{R} \rightarrow \mathbb{R}$ to make the example work. Find another instance of *GoodClass*.

```

class GoodClass t where
  constF ::  $\mathbb{R} \rightarrow t$ 
  addF ::  $t \rightarrow t \rightarrow t$ 
  mulF ::  $t \rightarrow t \rightarrow t$ 
  expF ::  $t \rightarrow t$ 
  -- ... Exercise: continue to mimic the FunExp datatype as a class
newtype FD a = FD (a  $\rightarrow$  a, a  $\rightarrow$  a)
instance Num a  $\Rightarrow$  GoodClass (FD a) where
  addF = evalDApp
  mulF = evalDMul
  expF = evalDExp
  -- ... Exercise: fill in the rest
  evalDApp = error "Exercise"
  evalDMul = error "Exercise"
  evalDExp = error "Exercise"

```

We can always define a homomorphism from *FunExp* to *any* instance of *GoodClass*, in an essentially unique way. In the language of category theory, the datatype *FunExp* is an initial algebra.

Let us explore this in the simpler context of *Monoid*. The language of monoids is given by

```

type Var    = String
data MExpr = Unit | Op MExpr MExpr | V Var

```

Alternatively, we could have parametrised *MExpr* over the type of variables.

Just as in the case of FOL terms, we can evaluate an *MExpr* in a monoid instance if we are given a way of interpreting variables, also called an assignment:

$$\text{evalM} :: \text{Monoid } a \Rightarrow (\text{Var} \rightarrow a) \rightarrow (\text{MExpr} \rightarrow a)$$

Once given an $f :: \text{Var} \rightarrow a$, the homomorphism condition defines *evalM*:

```

evalM f Unit      = unit
evalM f (Op e1 e2) = op (evalM f e1) (evalM f e2)
evalM f (V x)     = f x

```

(Observation: In *FunExp*, the role of variables was played by \mathbb{R} , and the role of the assignment by the identity.)

The following correspondence summarises the discussion so far:

| Computer Science | Mathematics |
|---------------------------------|---------------------------------------|
| DSL | structure (category, algebra, ...) |
| deep embedding, abstract syntax | initial algebra |
| shallow embedding | any other algebra |
| semantics | homomorphism from the initial algebra |

The underlying theory of this table is a fascinating topic but mostly out of scope for these lecture notes (and the DSLsofMath course). See Category Theory and Functional Programming for a whole course around this (lecture notes are available on github).

4.2.3 Other homomorphisms

In Section 3.6.1, we defined a *Num* instance for functions with a *Num* codomain. If we have an element of the domain of such a function, we can use it to obtain a homomorphism from functions to their codomains:

$$\text{Num } a \Rightarrow x \rightarrow (x \rightarrow a) \rightarrow a$$

As suggested by the type, the homomorphism is just function application:

$$\begin{aligned} \text{apply} &:: a \rightarrow (a \rightarrow b) \rightarrow b \\ \text{apply } a &= \lambda f \rightarrow f \ a \end{aligned}$$

Indeed, writing $h = \text{apply } c$ for some fixed c , we have

$$\begin{aligned} h (f + g) &= \{- \text{ def. } \text{apply } - \} \\ (f + g) \ c &= \{- \text{ def. } + \text{ for functions } - \} \\ f \ c + g \ c &= \{- \text{ def. } \text{apply } - \} \\ h \ f + h \ g & \end{aligned}$$

etc.

Can we do something similar for *FD*?

The elements of $FD \ a$ are pairs of functions, so we can take

$$\begin{aligned} \text{applyFD} &:: a \rightarrow FD \ a \rightarrow (a, a) \\ \text{applyFD } c \quad (FD \ (f, f')) &= FD \ (f \ c, f' \ c) \end{aligned}$$

We now have the domain of the homomorphism ($FD \ a$) and the homomorphism itself ($\text{applyFD } c$), but we are missing the structure on the codomain, which now consists of pairs (a, a) . In fact, we can *compute* this structure from the homomorphism condition. For example (we skip the constructor *FD* for brevity):

$$\begin{aligned} h ((f, f') * (g, g')) &= \{- \text{ def. } * \text{ for } FD \ a \ - \} \\ h (f * g, f' * g + f * g') &= \{- \text{ def. } h = \text{applyFD } c \ - \} \\ ((f * g) \ c, (f' * g + f * g') \ c) &= \{- \text{ def. } * \text{ and } + \text{ for functions } - \} \\ (f \ c * g \ c, f' \ c * g \ c + f \ c * g' \ c) &= \{- \text{ homomorphism condition from step 1 } - \} \\ h (f, f') \otimes h (g, g') &= \{- \text{ def. } h = \text{applyFD } c \ - \} \\ (f \ c, f' \ c) \otimes (g \ c, g' \ c) & \end{aligned}$$

The identity will hold if we take

$$\begin{aligned} \text{type } Dup \ a &= (a, a) \\ (\otimes) &:: Num \ a \Rightarrow Dup \ a \rightarrow Dup \ a \rightarrow Dup \ a \\ (x, x') \otimes (y, y') &= (x * y, x' * y + x * y') \end{aligned}$$

Thus, if we define a “multiplication” on pairs of values using (\otimes) , we get that $(\text{applyFD } c)$ is a *Num*-homomorphism for all c (or, at least for the operation $(*)$). We can now define an instance

$$\begin{aligned} \text{instance } Num \ a \Rightarrow Num \ (Dup \ a) \text{ where} \\ (*) &= (\otimes) \\ &\text{-- ... exercise} \end{aligned}$$

Exercise: complete the instance declarations for (\mathbb{R}, \mathbb{R}) .

Note: As this computation goes through also for the other cases we can actually work with just pairs of values (at an implicit point $c :: a$) instead of pairs of functions. Thus we can define a variant of *FD* a to be **type** *Dup* $a = (a, a)$

Hint: Something very similar can be used for Assignment 2.

4.3 Summing up: definitions and representation

We defined a *Num* structure on pairs (\mathbb{R}, \mathbb{R}) by requiring the operations to be compatible with the interpretation $(f\ a, f'\ a)$. For example

$$(x, x') \otimes (y, y') = (x * y, x' * y + x * y')$$

There is nothing in the “nature” of pairs of \mathbb{R} that forces this definition upon us. We chose it, because of the intended interpretation.

This multiplication is obviously not the one we need for *complex numbers*:

$$(x, x') * (y, y') = (x * y - x' * y', x * y' + x' * y)$$

Again, there is nothing in the nature of pairs that foists this operation on us. In particular, it is, strictly speaking, incorrect to say that a complex number *is* a pair of real numbers. The correct interpretation is that a complex number can be *represented* by a pair of real numbers, provided we define the operations on these pairs in a suitable way.

The distinction between definition and representation is similar to the one between specification and implementation, and, in a certain sense, to the one between syntax and semantics. All these distinctions are frequently obscured, for example, because of prototyping (working with representations / implementations / concrete objects in order to find out what definition / specification / syntax is most adequate). They can also be context-dependent (one man’s specification is another man’s implementation). Insisting on the difference between definition and representation can also appear quite pedantic (as in the discussion of complex numbers above). In general though, it is a good idea to be aware of these distinctions, even if they are suppressed for reasons of brevity or style. We will see this distinction again in section 5.1.

4.3.1 Some helper functions

```
instance Num E where -- Some abuse of notation (no proper negate, etc.)
  (+) = Add
  (*) = Mul
  fromInteger = Con
  negate = negateE
  negateE (Con c) = Con (negate c)
  negateE _ = error "negate: not supported"
```

4.4 Co-algebra and the Stream calculus

In the coming chapters there will be quite a bit of material on infinite structures. These are often captured not by algebras, but by co-algebras. We will not build up a general theory of co-algebras in these notes, but I could not resist to introduce a few examples which hint at the important role co-algebra plays in calculus.

Streams as an abstract datatype. Consider the API for streams of values of type A represented by some abstract type X :

```

data X
data A
head  :: X → A
tail  :: X → X
cons  :: A → X → X

law1  s =      s == cons (head s) (tail s)
law2 a s =      s == tail (cons a s)
law3 a s =      a == head (cons a s)

```

With this API we can use *head* to extract the first element of the stream, and *tail* to extract the rest as a new stream of type X . Using *head* and *tail* recursively we can extract an infinite list of values of type A :

```

toList :: X → [A]
toList x = head x : toList (tail x)

```

In the other direction, if we want to build a stream we only have one constructor: *cons* but no “base case”. In Haskell, thanks to laziness, we can still define streams directly using *cons* and recursion. As an example, we can construct a constant stream as follows:

```

constS :: A → X
constS a = ca
  where ca = cons a ca

```

Instead of specifying a stream in terms of how to construct it, we could describe it in terms of how to take it apart; by specifying its *head* and *tail*. In the constant stream example we would get something like:

```

head (constS a) = a
tail (constS a) = constS a

```

but this syntax is not supported in Haskell.

The last part of the API are a few laws we expect to hold. The first law simply states that if we first take a stream s apart into its head and its tail, we can get back to the original stream by *consing* them back together. The second and third are variant on this theme, and together the three laws specify how the three operations interact.

An unusual stream (Credits: [Pavlovic and Escardó, 1998])

Now consider $X = \mathbb{R} \rightarrow \mathbb{R}$, and $A = \mathbb{R}$ with the following definitions:

```

type X = ℝ → ℝ
type A = ℝ
deriv :: X → X
integ :: X → X
head f = f 0           -- value of  $f$  at 0
tail f = deriv f       -- derivative of  $f$ 
cons a f = const a + integ f -- start at  $a$ , integrate  $f$  from 0

```

Then the first law becomes

```

law1c f =
  f == cons (head f) (tail f)
    == (head f) + integ (tail f)
    == f 0 + integ (deriv f)

```

or, in traditional notation:

$$f(x) = f(0) + \int_0^x f'(t)dt$$

which we recognize as the fundamental law of calculus! There is much more to discover in this direction and we present some of it in the next few chapters.

For the curious. Here are the other two stream laws, in case you wondered.

```

law2c a f =
  f == tail (cons a f)
    == deriv (const a + integ f)
    == deriv (integ f)

```

```

law3c a f =
  a == head (cons a f)
  a == head (const a + integ f)
  a == (const a + integ f) 0
  a == a + (integ f) 0
  0 == integ f 0

```

4.5 Exercises

Exercise 4.1. Homomorphisms. Consider the following definitions:

```

-- h : A → B is a homomorphism from Op : A → A → A to op : B → B → B
H2 (h, Op, op) = ∀ x. ∀ y. h (Op x y) == op (h x) (h y)
-- h : A → B is a homomorphism from F : A → A to f : B → B
H1 (h, F, f)   = ∀ x. h (F x) == f (h x)
-- h : A → B is a homomorphism from E : A to e : B
H0 (h, E, e)   = h E == e

```

Prove or disprove the following claims:

- $H_2 ((2*), (+), (+))$
- $H_2 ((2*), (*), (*))$
- $H_2 (exp, (+), (*))$
- $H_2 (eval', (:+), (+))$
- $H_1 (\sqrt{\cdot}, (4*), (2*))$
- $\exists f. H_1 (f, (2*) \circ (1+), (1+) \circ (2*))$

Exercise 4.2. Complete the instance declarations for *FunExp* (for *Num*, *Fractional*, and *Floating*).

Exercise 4.3. Complete the instance declarations for (\mathbb{R}, \mathbb{R}) , deriving them from the homomorphism requirement for *apply* (in section 4.2.3).

Exercise 4.4. We now have three different ways of computing the derivative of a function such as $f\ x = \sin x + \exp (\exp x)$ at a given point, say $x = \pi$.

- Find $e :: \text{FunExp}$ such that $\text{eval } e = f$ and use *eval'*.
- Find an expression of type $\text{FD } \mathbb{R}$ and use *apply*.
- Apply f directly to the appropriate (x, x') and use *snd*.

Do you get the same result?

Exercise 4.5. *From exam 2017-08-22*

In exercise 1.3 we looked at the datatype $\text{SR } v$ for the language of semiring expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *SemiRing* that corresponds to the semiring structure.
- Define a *SemiRing* instance for the datatype $\text{SR } v$ that you defined in exercise 1.3.
- Find two other instances of the *SemiRing* class.
- Specialise the evaluator that you defined in exercise 1.3 to the two *SemiRing* instances defined above. Take three semiring expressions of type SR String , give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.6. Show that arithmetic modulo n satisfies the semiring laws (it is even a ring). In more details: show that $\mathbb{Z}_n = \{0, 1, \dots, n-1\}$ with *plus* $x\ y = (x+y)\%n$ and *times* $x\ y = (x*y)\%n$ forms a semiring.

With $h\ x = x\%n$, show that h is a homomorphism from \mathbb{Z} to \mathbb{Z}_n .

Exercise 4.7. *From exam 2016-03-15*

In exercise 1.4, we looked a datatype for the language of lattice expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *Lattice* that corresponds to the lattice structure.
- Define a *Lattice* instance for the datatype for lattice expressions that you defined in 1.4.1.
- Find two other instances of the *Lattice* class.
- Specialise the evaluator you defined in exercise 1.4.2 to the two *Lattice* instances defined above. Take three lattice expressions, give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.8. *From exam 2016-08-23*

In exercise 1.5, we looked a datatype for the language of abelian monoid expressions. We will now use some of the concepts discussed in this chapter to expand on this language.

- Define a type class *AbMonoid* that corresponds to the abelian monoid structure.
- Define an *AbMonoid* instance for the datatype for abelian monoid expressions that you defined in exercise 1.5.1.

- c. Find one other instance of the *AbMonoid* class and give an example which is **not** an instance of *AbMonoid*.
- d. Specialise the evaluator that you defined in exercise 1.5.2 to the *AbMonoid* instance defined above. Take three ‘AbMonoidExp’ expressions, give the appropriate assignments and compute the results of evaluating the three expressions.

Exercise 4.9. (Closely related to exam question)

A *ring* is a set A together with two constants (or nullary operations), 0 and 1, one unary operation, *negate*, and two binary operations, $+$ and $*$, such that

- a. 0 is the neutral element of $+$

$$\forall x \in A. \ x + 0 = 0 + x = x$$

- b. $+$ is associative

$$\forall x, y, z \in A. \ x + (y + z) = (x + y) + z$$

- c. *negate* inverts elements with respect to addition

$$\forall x \in A. \ x + \text{negate } x = \text{negate } x + x = 0$$

- d. $+$ is commutative

$$\forall x, y \in A. \ x + y = y + x$$

- e. 1 is the unit of $*$

$$\forall x \in A. \ x * 1 = 1 * x = x$$

- f. $*$ is associative

$$\forall x, y, z \in A. \ x * (y * z) = (x * y) * z$$

- g. $*$ distributes over $+$

$$\begin{aligned} \forall x, y, z \in A. \ x * (y + z) &= (x * y) + (x * z) \\ \forall x, y, z \in A. \ (x + y) * z &= (x * z) + (y * z) \end{aligned}$$

Remarks:

- a. and b. say that $(A, 0, +)$ is a monoid
- a–c. say that $(A, 0, +, \text{negate})$ is a group
- a–d. say that $(A, 0, +, \text{negate})$ is a commutative group
- e. and f. say that $(A, 1, *)$ is a monoid

- i Define a type class *Ring* that corresponds to the ring structure.

- ii Define a datatype for the language of ring expressions (including variables) and define a *Ring* instance for it.
- iii Find two other instances of the *Ring* class.
- iv Define a general evaluator for *Ring* expressions on the basis of a given assignment function.
- v Specialise the evaluator to the two *Ring* instances defined at point iii. Take three ring expressions, give the appropriate assignments and compute the results of evaluating, in each case, the three expressions.

Exercise 4.10. *From exam 2017-03-14*

Recall the type of expressions

```
data FunExp = Const Rational      | Id
             | FunExp :+: FunExp | Exp FunExp
             | FunExp :*: FunExp | Sin FunExp
             | FunExp :/: FunExp | Cos FunExp
             -- and so on
deriving Show
```

and consider the function

```
f :: ℝ → ℝ
f x = exp (sin x) + x
```

- a. Find an expression e such that $\text{eval } e == f$ and show this using equational reasoning.
- b. Implement a function *deriv2* such that, for any $f : \text{Fractional } a \Rightarrow a \rightarrow a$ constructed with the grammar of *FunExp* and any x in the domain of f , we have that $\text{deriv2 } f \ x$ computes the second derivative of f at x . Use the function $\text{derive} :: \text{FunExp} \rightarrow \text{FunExp}$ from the lectures ($\text{eval } (\text{derive } e)$ is the derivative of $\text{eval } e$). What instance declarations do you need?

The type of $\text{deriv2 } f$ should be $\text{Fractional } a \Rightarrow a \rightarrow a$.

Exercise 4.11. Based on the lecture notes, complete all the instance and datatype declarations and definitions in the files *FunNumInst.lhs*, *FunExp.lhs*, *Derive.lhs*, *EvalD.lhs*, and *ShallowD.lhs*.

Exercise 4.12. Write a function

```
simplify :: FunExp → FunExp
```

to simplify the expression resulted from *derive*. For example, the following tests should work:

```
simplify (Const 0 :*: Exp Id) == Const 0
simplify (Const 0 :+: Exp Id) == Exp Id
simplify (Const 2 :*: Const 1) == Const 2
simplify (derive (Id :*: Id))  == Const 2 :*: Id
```

As a motivating example, note that $\text{derive } (\text{Id} :*: \text{Id})$ evaluates to $(\text{Const } 1.0 :*: \text{Id}) :+: (\text{Id} :*: \text{Const } 1.0)$ without *simplify*, and that the second derivative looks even worse.

